<commit_before>@ECHO OFF

REM This assumes NUGET.EXE is in the same directory as this script or in the path
REM If not download it from  http://nuget.org/nuget.exe

SETLOCAL

REM Does string have a trailing slash? if so remove it
SET THISDIR=%~dp0
IF %THISDIR:~-1%==\ SET THISDIR=%THISDIR:~0,-1%

SET WORKING=%THISDIR%\working
SET TOOLS=%WORKING%\tools
SET SRC=%THISDIR%\..\src

PUSHD
CD /D %THISDIR%

ECHO Cleaning the working directory
RD /S /Q "%WORKING%" > NUL
MKDIR "%WORKING%"

ECHO Copy source ready for packing...
MKDIR "%TOOLS%"
XCOPY "%SRC%" "%TOOLS%" /s /e /v /c /i /f /y

ECHO Removing extra files...
DEL "%WORKING%\Tools\InstallPowerYamlGithub.cmd"

SET PKGVERSION=%APPVEYOR_BUILD_VERSION%
IF NOT [%PKGVERSION%] == [] SET PKGVERSION=-Version "0.9.%PKGVERSION%"
ECHO Using package version %PKGVERSION%

ECHO Run Nuget Pack
"nuget.exe" pack  "%~dp0Package.nuspec" -OutputDirectory "%THISDIR%" -BasePath "%WORKING%" -NonInteractive %PKGVERSION%

EXIT /B %ERRORLEVEL%<commit_msg>Updated build.bat to build a functional chocolatey package<commit_after>@ECHO OFF

REM This assumes NUGET.EXE is in the same directory as this script or in the path
REM If not download it from  http://nuget.org/nuget.exe

SETLOCAL

REM Does string have a trailing slash? if so remove it
SET THISDIR=%~dp0
IF %THISDIR:~-1%==\ SET THISDIR=%THISDIR:~0,-1%

SET WORKING=%THISDIR%\working
SET TOOLS=%WORKING%\tools
SET TOOLS_MODULE=%WORKING%\tools\module
SET SRC=%THISDIR%\..\src

PUSHD
CD /D %THISDIR%

ECHO Cleaning the working directory
RD /S /Q "%WORKING%" > NUL
MKDIR "%WORKING%"
ECHO Creating directory structure
MKDIR "%TOOLS%"
MKDIR "%TOOLS_MODULE%"

ECHO Copy source ready for packing...
XCOPY "%SRC%" "%TOOLS_MODULE%" /s /e /v /c /i /f /y

ECHO Removing extra files...
DEL "%TOOLS_MODULE%\InstallPowerYamlGithub.cmd"

ECHO Copying chocolatey support files...
COPY "%THISDIR%\chocolateyInstall.ps1" "%TOOLS%\chocolateyInstall.ps1"
COPY "%THISDIR%\chocolateyUninstall.ps1" "%TOOLS%\chocolateyUninstall.ps1"

ECHO Getting the package version from various sources
SET PKGVERSION=0.0.1
IF NOT [%1] == [] SET PKGVERSION=%1
IF NOT [%APPVEYOR_BUILD_VERSION%] == [] SET PKGVERSION=0.9.%APPVEYOR_BUILD_VERSION%
ECHO Using package version %PKGVERSION%
IF NOT [%PKGVERSION%] == [] SET PKGVERSION=-Version "%PKGVERSION%"

ECHO Run Nuget Pack
COPY "%THISDIR%\Package.nuspec" "%WORKING%\Package.nuspec"
"nuget.exe" pack  "%WORKING%\Package.nuspec" -NonInteractive %PKGVERSION% -NoPackageAnalysis

EXIT /B %ERRORLEVEL%
{'fraction_non_alphanumeric': 0.13306613226452907, 'fraction_numerical': 0.008817635270541082, 'mean_word_length': 4.92874109263658, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 5, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 2, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '4284007', 'n_tokens_mistral': 980, 'n_tokens_neox': 887, 'n_words': 279}